

NAG Fortran Library Routine Document

F08YLF (DTGSNA)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08YLF (DTGSNA) estimates condition numbers for specified eigenvalues and/or eigenvectors of a matrix pair in generalized real Schur form.

2 Specification

```

SUBROUTINE F08YLF (JOB, HOWMNY, SELECT, N, A, LDA, B, LDB, VL, LDVL, VR,
1 LDVR, S, DIF, MM, M, WORK, LWORK, IWORK, INFO)
INTEGER N, LDA, LDB, LDVL, LDVR, MM, M, LWORK, IWORK(*), INFO
double precision A(LDA,*), B(LDB,*), VL(LDVL,*), VR(LDVR,*), S(*),
1 DIF(*), WORK(*)
LOGICAL SELECT(*)
CHARACTER*1 JOB, HOWMNY

```

The routine may be called by its LAPACK name *dtgsna*.

3 Description

F08YLF (DTGSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of an n by n matrix pair (S, T) in real generalized Schur form. The routine actually returns estimates of the reciprocals of the condition numbers in order to avoid possible overflow.

The pair (S, T) are in real generalized Schur form if S is block upper triangular with 1 by 1 and 2 by 2 diagonal blocks and T is upper triangular as returned, for example, by F08XAF (DGGES), or F08XEF (DHGEQZ) with $JOB = 'S'$. The diagonal elements, or blocks, define the generalized eigenvalues (α_i, β_i) , for $i = 1, 2, \dots, n$ of the pair (S, T) and the eigenvalues are given by

$$\lambda_i = \alpha_i / \beta_i,$$

so that

$$\beta_i S x_i = \alpha_i T x_i \quad \text{or} \quad S x_i = \lambda_i T x_i,$$

where x_i is the corresponding (right) eigenvector.

If S and T are the result of a generalized Schur factorization of a matrix pair (A, B)

$$A = QSZ^T, \quad B = QTZ^T$$

then the eigenvalues and condition numbers of the pair (S, T) are the same as those of the pair (A, B) .

Let $(\alpha, \beta) \neq (0, 0)$ be a simple generalized eigenvalue of (A, B) . Then the reciprocal of the condition number of the eigenvalue $\lambda = \alpha / \beta$ is defined as

$$s(\lambda) = \frac{\left(|y^T A x|^2 + |y^T B x|^2 \right)^{1/2}}{(\|x\|_2 \|y\|_2)},$$

where x and y are the right and left eigenvectors of (A, B) corresponding to λ . If both α and β are zero, then (A, B) is singular and $s(\lambda) = -1$ is returned.

The definition of the reciprocal of the estimated condition number of the right eigenvector x and the left eigenvector y corresponding to the simple eigenvalue λ depends upon whether λ is a real eigenvalue, or one of a complex conjugate pair.

If the eigenvalue λ is real and U and V are orthogonal transformations such that

$$U^T(A, B)V = (S, T) = \begin{pmatrix} \alpha & * \\ 0 & S_{22} \end{pmatrix} \begin{pmatrix} \beta & * \\ 0 & T_{22} \end{pmatrix},$$

where S_{22} and T_{22} are $(n-1)$ by $(n-1)$ matrices, then the reciprocal condition number is given by

$$\text{Dif}(x) \equiv \text{Dif}(y) = \text{Dif}((\alpha, \beta), (S_{22}, T_{22})) = \sigma_{\min}(Z),$$

where $\sigma_{\min}(Z)$ denotes the smallest singular value of the $2(n-1)$ by $2(n-1)$ matrix

$$Z = \begin{pmatrix} \alpha \otimes I & -1 \otimes S_{22} \\ \beta \otimes I & -1 \otimes T_{22} \end{pmatrix}$$

and \otimes is the Kronecker product.

If λ is part of a complex conjugate pair and U and V are orthogonal transformations such that

$$U^T(A, B)V = (S, T) = \begin{pmatrix} S_{11} & * \\ 0 & S_{22} \end{pmatrix} \begin{pmatrix} T_{11} & * \\ 0 & T_{22} \end{pmatrix},$$

where S_{11} and T_{11} are two by two matrices, S_{22} and T_{22} are $(n-2)$ by $(n-2)$ matrices, and (S_{11}, T_{11}) corresponds to the complex conjugate eigenvalue pair $\lambda, \bar{\lambda}$, then there exist unitary matrices U_1 and V_1 such that

$$U_1^H S_{11} V_1 = \begin{pmatrix} s_{11} & s_{12} \\ 0 & s_{22} \end{pmatrix} \quad \text{and} \quad U_1^H T_{11} V_1 = \begin{pmatrix} t_{11} & t_{12} \\ 0 & t_{22} \end{pmatrix}.$$

The eigenvalues are given by $\lambda = s_{11}/t_{11}$ and $\bar{\lambda} = s_{22}/t_{22}$. Then the Frobenius norm-based, estimated reciprocal condition number is bounded by

$$\text{Dif}(x) \equiv \text{Dif}(y) \leq \min(d_1, \max(1, |\text{Re}(s_{11})/\text{Re}(s_{22})|), d_2)$$

where $\text{Re}(z)$ denotes the real part of z , $d_1 = \text{Dif}((s_{11}, t_{11}), (s_{22}, t_{22})) = \sigma_{\min}(Z_1)$, Z_1 is the complex two by two matrix

$$Z_1 = \begin{pmatrix} s_{11} & -s_{22} \\ t_{11} & -t_{22} \end{pmatrix},$$

and d_2 is an upper bound on $\text{Dif}((S_{11}, T_{11}), (S_{22}, T_{22}))$; that is, an upper bound on $\sigma_{\min}(Z_2)$, where Z_2 is the $(2n-2)$ by $(2n-2)$ matrix

$$Z_2 = \begin{pmatrix} S_{11}^T \otimes I & -I \otimes S_{22} \\ T_{11}^T \otimes I & -I \otimes T_{22} \end{pmatrix}.$$

See Anderson *et al.* (1999) (Sections 2.4.8 and 4.11) and Kågström and Poromaa (1996) for further details and information.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Kågström B and Poromaa P (1996) LAPACK-style algorithms and software for solving the generalized Sylvester equation and estimating the separation between regular matrix pairs *ACM Trans. Math. Software* **22** 78–103

5 Parameters

1: JOB – CHARACTER*1

Input

On entry: indicates whether condition numbers are required for eigenvalues and/or eigenvectors.

JOB = 'E'

Condition numbers for eigenvalues only are computed.

JOB = 'V'

Condition numbers for eigenvectors only are computed.

JOB = 'B'

Condition numbers for both eigenvalues and eigenvectors are computed.

Constraint: JOB = 'E', 'V' or 'B'.

2: HOWMNY – CHARACTER*1 *Input*

On entry: indicates how many condition numbers are to be computed.

HOWMNY = 'A'

Condition numbers for all eigenpairs are computed.

HOWMNY = 'S'

Condition numbers for selected eigenpairs (as specified by SELECT) are computed.

Constraint: HOWMNY = 'A' or 'S'.

3: SELECT(*) – LOGICAL array *Input*

Note: the dimension of the array SELECT must be at least $\max(1, N)$ if HOWMNY = 'S' and at least 1 otherwise.

On entry: specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the real eigenvalue λ_j , SELECT(j) must be set .TRUE.. To select condition numbers corresponding to a complex conjugate pair of eigenvalues λ_j and λ_{j+1} , SELECT(j) and/or SELECT($j + 1$) must be set to .TRUE..

If HOWMNY = 'A', SELECT is not referenced.

4: N – INTEGER *Input*

On entry: n , the order of the matrix pair (S, T) .

Constraint: $N \geq 0$.

5: A(LDA,*) – **double precision** array *Input*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the upper quasi-triangular matrix S .

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08YLF (DTGSNA) is called.

Constraint: $LDA \geq \max(1, N)$.

7: B(LDB,*) – **double precision** array *Input*

Note: the second dimension of the array B must be at least $\max(1, N)$.

On entry: the upper triangular matrix T .

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08YLF (DTGSNA) is called.

Constraint: $LDB \geq \max(1, N)$.

- 9: VL(LDVL,*) – *double precision* array *Input*
Note: the second dimension of the array VL must be at least $\max(1, M)$.
On entry: if JOB = 'E' or 'B', VL must contain left eigenvectors of (S, T) , corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors must be stored in consecutive columns of VL, as returned by F08YKF (DTGEVC) or F08WAF (DGGEV).
 If JOB = 'V', VL is not referenced.
- 10: LDVL – INTEGER *Input*
On entry: the first dimension of the array VL as declared in the (sub)program from which F08YLF (DTGSNA) is called.
Constraints:
 if JOB = 'E' or 'B', $LDVL \geq N$;
 $LDVL \geq 1$ otherwise.
- 11: VR(LDVR,*) – *double precision* array *Input*
Note: the second dimension of the array VR must be at least $\max(1, M)$.
On entry: if JOB = 'E' or 'B', VR must contain right eigenvectors of (S, T) , corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors must be stored in consecutive columns of VR, as returned by F08YKF (DTGEVC) or F08WAF (DGGEV).
 If JOB = 'V', VR is not referenced.
- 12: LDVR – INTEGER *Input*
On entry: the first dimension of the array VR as declared in the (sub)program from which F08YLF (DTGSNA) is called.
Constraints:
 if JOB = 'E' or 'B', $LDVR \geq N$;
 $LDVR \geq 1$ otherwise.
- 13: S(*) – *double precision* array *Output*
Note: the dimension of the array S must be at least $\max(1, MM)$.
On exit: if JOB = 'E' or 'B', the reciprocal condition numbers of the selected eigenvalues, stored in consecutive elements of the array. For a complex conjugate pair of eigenvalues two consecutive elements of S are set to the same value. Thus $S(j)$, $DIF(j)$, and the j th columns of VL and VR all correspond to the same eigenpair (but not in general the j th eigenpair, unless all eigenpairs are selected).
 If JOB = 'V', S is not referenced.
- 14: DIF(*) – *double precision* array *Output*
Note: the dimension of the array DIF must be at least $\max(1, MM)$.
On exit: if JOB = 'V' or 'B', the estimated reciprocal condition numbers of the selected eigenvectors, stored in consecutive elements of the array. For a complex eigenvector two consecutive elements of DIF are set to the same value. If the eigenvalues cannot be reordered to compute $DIF(j)$, $DIF(j)$ is set to 0; this can only occur when the true value would be very small anyway.
 If JOB = 'E', DIF is not referenced.
- 15: MM – INTEGER *Input*
On entry: the number of elements in the arrays S and DIF.
Constraint: $MM \geq M$.

- 16: M – INTEGER *Output*
On exit: the number of elements of the arrays S and DIF used to store the specified condition numbers; for each selected real eigenvalue one element is used, and for each selected complex conjugate pair of eigenvalues, two elements are used. If HOWMNY = 'A', M is set to N.
- 17: WORK(*) – *double precision* array *Workspace*
Note: the dimension of the array WORK must be at least max(1, LWORK).
On exit: if INFO = 0, WORK(1) returns the minimum LWORK.
- 18: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08YLF (DTGSNA) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the minimum size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
Constraints:
 if LWORK \neq -1,
 if JOB = 'V' or 'B', LWORK $\geq 2 \times N \times (N + 2) + 16$;
 LWORK \geq max(1, N) otherwise.
- 19: IWORK(*) – INTEGER array *Workspace*
Note: the dimension of the array IWORK must be at least (N + 6).
 If JOB = 'E', IWORK is not referenced.
- 20: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = -i, the i-th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

None.

8 Further Comments

An approximate asymptotic error bound on the chordal distance between the computed eigenvalue $\tilde{\lambda}$ and the corresponding exact eigenvalue λ is

$$\chi(\tilde{\lambda}, \lambda) \leq \epsilon \|(A, B)\|_F / S(\lambda)$$

where ϵ is the *machine precision*.

An approximate asymptotic error bound for the right or left computed eigenvectors \tilde{x} or \tilde{y} corresponding to the right and left eigenvectors x and y is given by

$$\theta(\tilde{z}, z) \leq \epsilon \|(A, B)\|_F / \text{Dif}.$$

The complex analogue of this routine is F08YYF (ZTGSNA).

9 Example

To estimate condition numbers and approximate error estimates for all the eigenvalues and eigenvalues and right eigenvectors of the pair (S, T) given by

$$S = \begin{pmatrix} 4.0 & 1.0 & 1.0 & 2.0 \\ 0 & 3.0 & -1.0 & 1.0 \\ 0 & 1.0 & 3.0 & 1.0 \\ 0 & 0 & 0 & 6.0 \end{pmatrix} \quad \text{and} \quad T = \begin{pmatrix} 2.0 & 1.0 & 1.0 & 3.0 \\ 0 & 1.0 & 0.0 & 1.0 \\ 0 & 0 & 1.0 & 1.0 \\ 0 & 0 & 0 & 2.0 \end{pmatrix}.$$

The eigenvalues and eigenvectors are computed by calling F08YKF (DTGEVC).

9.1 Program Text

```
*      F08YLF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER       (NMAX=8)
      INTEGER          LDS, LDT, LDVL, LDVR, LWORK
      PARAMETER       (LDS=NMAX,LDT=NMAX,LDVL=NMAX,LDVR=NMAX,
+                    LWORK=2*NMAX*(NMAX+2)+16)
*      .. Local Scalars ..
      DOUBLE PRECISION EPS, SNORM, STNRM, TNORM
      INTEGER          I, INFO, J, M, N
*      .. Local Arrays ..
      DOUBLE PRECISION DIF(NMAX), S(LDS,NMAX), SCON(NMAX), T(LDT,NMAX),
+                    VL(LDVL,NMAX), VR(LDVR,NMAX), WORK(LWORK)
      INTEGER          IWORK(NMAX+6)
      LOGICAL          SELECT(1)
*      .. External Functions ..
      DOUBLE PRECISION F06BNF, F06RAF, X02AJF
      EXTERNAL         F06BNF, F06RAF, X02AJF
*      .. External Subroutines ..
      EXTERNAL         DTGEVC, DTGSNA
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08YLF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*         Read S and T from data file
*
*
      READ (NIN,*) ((S(I,J),J=1,N),I=1,N)
      READ (NIN,*) ((T(I,J),J=1,N),I=1,N)
*
*         Calculate the left and right generalized eigenvectors of the
*         pair (S,T). Note that DTGEVC requires WORK to be of dimension
*         at least 6*N, and LWORK is always greater than 6*NMAX.
*
      CALL DTGEVC('Both','All',SELECT,N,S,LDS,T,LDT,VL,LDVL,VR,LDVR,
+              N,M,WORK,INFO)
      IF (INFO.GT.0) THEN
          WRITE (NOUT,99999) INFO, INFO + 1
          STOP
      END IF
*
*         Estimate condition numbers for all the generalized eigenvalues
*         and right eigenvectors of the pair (S,T)
*
      CALL DTGSNA('Both','All',SELECT,N,S,LDS,T,LDT,VL,LDVL,VR,LDVR,
+              SCON,DIF,N,M,WORK,LWORK,IWORK,INFO)
*
*         Print condition numbers of eigenvalues and right eigenvectors
*

```

```

      WRITE (NOUT,*) 'SCON'
      WRITE (NOUT,99998) (SCON(I),I=1,M)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'DIF'
      WRITE (NOUT,99998) (DIF(I),I=1,M)
*
*      Calculate approximate error estimates
*
*      Compute the 1-norms of S and T and then compute
*      SQRT(SNORM**2 + TNORM**2)
*
      EPS = X02AJF()
      SNORM = F06RAF('1-norm',N,N,S,LDS,WORK)
      TNORM = F06RAF('1-norm',N,N,T,LDT,WORK)
      STNRM = F06BNF(SNORM,TNORM)
      WRITE (NOUT,*)
      WRITE (NOUT,*)
+      'Approximate error estimates for eigenvalues of (S,T)'
      WRITE (NOUT,99998) (EPS*STNRM/SCON(I),I=1,M)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Approximate error estimates for right ',
+      'eigenvectors of (S,T)'
      WRITE (NOUT,99998) (EPS*STNRM/DIF(I),I=1,M)
      ELSE
      WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (' The 2-by-2 block (' ,I5,':',I5,') does not have a', ' co',
+      'mplex eigenvalue')
99998 FORMAT ((3X,1P,7E11.1))
      END

```

9.2 Program Data

```

F08YLF Example Program Data
  4      :Value of N
  4.0  1.0  1.0  2.0
  0.0  3.0 -1.0  1.0
  0.0  1.0  3.0  1.0
  0.0  0.0  0.0  6.0  :End of matrix S
  2.0  1.0  1.0  3.0
  0.0  1.0  0.0  1.0
  0.0  0.0  1.0  1.0
  0.0  0.0  0.0  2.0  :End of matrix T

```

9.3 Program Results

```

F08YLF Example Program Results

SCON
  1.6E+00    1.7E+00    1.7E+00    1.4E+00

DIF
  5.4E-01    1.5E-01    1.5E-01    1.2E-01

Approximate error estimates for eigenvalues of (S,T)
  8.7E-16    7.8E-16    7.8E-16    9.9E-16

Approximate error estimates for right eigenvectors of (S,T)
  2.5E-15    9.0E-15    9.0E-15    1.1E-14

```
